

Oxygen Toxicity Calculations

by Erik C. Baker, P.E.

Management of exposure to oxygen toxicity is an important element of technical diving. Most often this is accomplished during the dive planning stage through the use of a dive/decompression planning program on a computer. Having developed such a program myself using the good ole' FORTRAN programming language, I found that incorporating oxygen toxicity calculations was more involved than just a straightforward application of the established equations. Obtaining "exact" calculations during ascent or descent segments of a dive profile presented the main challenge. So I did some more research, dusted-off my calculus book, and came up with some calculation methods which I would like to share with the technical diving community. The information presented in this article is intended to serve as a primer or review for some and to offer programming suggestions for others.

Background

Two oxygen toxicity parameters are typically "tracked" in technical diving calculations. The first is pulmonary oxygen toxicity which primarily concerns the effects to the lungs of long-term exposures to oxygen at elevated partial pressures. The second is central nervous system (CNS) oxygen toxicity which primarily concerns the effects to the brain of short-to-medium-term exposures to oxygen at elevated partial pressures. Both oxygen toxicity parameters are a function of the partial pressure of oxygen (PO_2) in the inspired breathing gas and the time of exposure. CNS oxygen toxicity is usually the parameter of most concern and greatest impact in technical diving, however some "mega-dives" undertaken in the recent past (Bushmansgat, Zacatón, etc.) have pushed the limits of pulmonary oxygen toxicity as well.

The equation for calculating pulmonary oxygen toxicity was developed from empirically derived pulmonary oxygen tolerance curves such as represented in Figure 1. The curves were based on measurements of rate of decrease in vital capacity in "normal men" during oxygen breathing at elevated partial pressures. Vital capacity is the maximum volume of air (or gas) that a person can expel from his/her lungs after first filling the lungs to their maximum extent and then expiring to the maximum extent. The tolerance curves are hyperbolic with asymptotes of zero time and $PO_2 = 0.5$ which is the approximate threshold for detectable pulmonary effects. The curve representing a 4% decrement in vital capacity was used to develop technical diving exposure limits for pulmonary oxygen toxicity.




Figure 1

Along with the tolerance curves, Dr. Christian J. Lambertsen and colleagues at the Institute for Environmental Medicine, University of Pennsylvania, developed the current “tracking” method for pulmonary oxygen toxicity in the form of the unit pulmonary toxic dose (UPTD) and the cumulative pulmonary toxic dose (CPTD). The UPTD is more commonly referred to in the diving community as the oxygen toxicity unit (OTU). One UPTD (or OTU) is the degree of pulmonary oxygen toxicity produced by breathing 100% O₂ continuously at a pressure of 1 atmosphere absolute (ATA) for 1 minute. The CPTD calculation (see Equation 1 below) converts any continuous oxygen exposure (PO₂ above 0.5) and time combination to be expressed as UPTD’s (or OTU’s). The CPTD calculation is carried out for each segment of the dive profile and the results (expressed as OTU’s) are summed up to produce the total number of OTU’s for the dive. This number can then be compared against the daily and multi-day limits established by Dr. Bill Hamilton and colleagues in the NOAA Repetitive Excursions (REPEX) Procedures Report. These published limits have been widely adopted by the technical diving community.

The NOAA limits for CNS oxygen toxicity (normal, single exposure) are plotted in Figure 2. These limits are based on research by Drs. F.K. Butler and Edward D. Thalmann, U.S. Navy Experimental Diving Unit, and by Drs. Christian J. Lambertsen and Russell E. Peterson, Institute for Environmental Medicine, University of Pennsylvania. These limits take operational safety into consideration and have been widely adopted by the technical diving community. The common method in use to “track” CNS oxygen toxicity is to compute a “CNS fraction” for each segment of the dive profile and then sum up these results to produce a total CNS fraction for the dive. These fractions are often multiplied by 100 and expressed as a percentage (CNS %). A CNS fraction is calculated by taking the time spent at a given PO₂ and dividing by the NOAA time limit for that PO₂. When the CNS fractions from all segments of the dive profile add up to 1 (or 100%) then the overall limit for CNS oxygen toxicity has been reached for that dive.




Figure 2

Since the NOAA limits for CNS oxygen toxicity are simply given as discrete data points, and not by a continuous mathematical function, it is convenient to “connect the dots” with straight lines as represented in Figure 2. In this manner, it is possible to create a series of linear equations which describe the CNS time limits for various ranges of PO₂. As demonstrated later in this article, this manipulation facilitates the calculation process, particularly by computer program.

For calculation purposes, oxygen toxicity is not defined at or below a PO₂ of 0.5. Whenever the PO₂ goes above 0.5 during a dive profile, either at a constant depth or during ascent or descent at a constant rate, the two oxygen toxicity parameters can be calculated.

One subject area regarding oxygen toxicity that this article does not address is that of “extension of oxygen tolerance.” There is some limited data in the academic literature which indicates that the practice of intermittent oxygen breathing (i.e. 20 minutes on O₂, then 5 minute air break) can extend the limits of pulmonary oxygen tolerance by a factor of more than 2. Presumably, the CNS oxygen limits are extended by some factor as well. The problem is that, to my knowledge, a reliable method has not yet been developed to calculate the rate of recovery from oxygen exposure **during these periods of intermittent oxygen breathing**. This is an area for future research by the academicians. It is, however, a common practice in technical diving to calculate a decay in the CNS fraction **during surface intervals** using an exponential decay equation with a half-time of 90 minutes.

Pulmonary Oxygen Toxicity Calculations

Constant depth profile:

For a constant depth profile (in which the PO₂ remains constant), the cumulative pulmonary toxic dose (CPTD), expressed as oxygen toxicity units (OTU), is calculated by Equation 1:

$$OTU = t_x \cdot \left(\frac{0.5}{PO_2 - 0.5} \right)^{\frac{-5}{6}} \quad (\text{Eq. 1})$$

where t_x is the time of exposure, PO_2 is constant, and $\frac{-5}{6}$ is an exponent ≈ -0.8333 .

Ascent or descent profile at a constant rate:

For an ascent or descent profile at a constant rate (in which the PO₂ varies at a constant rate), the CPTD, expressed as OTU's, can be calculated by Equation 2:

$$OTU = \frac{\frac{3}{11} \cdot t_x}{PO_{2f} - PO_{2i}} \cdot \left[\left(\frac{PO_{2f} - 0.5}{0.5} \right)^{\frac{11}{6}} - \left(\frac{PO_{2i} - 0.5}{0.5} \right)^{\frac{11}{6}} \right] \quad (\text{Eq. 2})$$

where t_x is the time of exposure over the interval, PO_{2i} is the initial PO₂ of the interval, PO_{2f} is the final PO₂ of the interval, and $\frac{11}{6}$ is an exponent ≈ 1.8333 .

Computer format:

In the FORTRAN programming language, Equation 1 can be written as:

$$OTU = TIME * ((0.5 / (PO_2 - 0.5)) ** (-5.0 / 6.0))$$

while Equation 2 can be written as:

$$OTU = 3.0 / 11.0 * TIME / (PO_{2F} - PO_{2I}) * (((PO_{2F} - 0.5) / 0.5) ** (11.0 / 6.0) - ((PO_{2I} - 0.5) / 0.5) ** (11.0 / 6.0))$$

Discussion:

As mentioned above, Equation 1 was developed by Lambertsen and colleagues from the pulmonary oxygen tolerance curves. Equation 2 can be derived from Equation 1 as follows:

Since the PO_2 varies at a constant rate with respect to time during an ascent or descent at a constant rate, then

$$\frac{dPO_2}{dt} = k \quad \int_{PO_{2o}}^{PO_{2f}} dPO_2 = \int_{t_o}^{t_f} k dt \quad PO_{2f} = PO_{2o} + kt$$

Equation 1 can now be expressed as a differential with PO_2 as a function of time:

$$dOTU = dt_x \cdot \left(\frac{0.5}{PO_{2o} + kt - 0.5} \right)^{\frac{-5}{6}}$$

Setting $a = PO_{2o} - 0.5$, Equation 2 is obtained from the following integration:

$$\int_{OTU_o}^{OTU_f} dOTU = \int_{t_o}^{t_f} dt_x \cdot \left(\frac{0.5}{a + kt} \right)^{\frac{-5}{6}}$$

Calculation steps before using Equation 2 for ascent or descent profile:

A few steps must be taken before Equation 2 can be applied for calculation. The following calculations are based on diving at sea level and are expressed in units of American usage:

1. Determine segment time for the ascent or descent profile (at a constant rate):

$$\text{Segment Time} = (\text{Final Depth} - \text{Starting Depth}) \div \text{Rate}^* \text{ of ascent or descent}$$

* Note: descent rate should be expressed as positive number (i.e. +60 fsw/min) while ascent rate should be expressed as a negative number (i.e. -30 fsw/min).

2. Determine initial and final pressures of the profile in atmospheres absolute (ATA):

$$\text{Initial Pressure (ATA)} = (\text{Starting Depth} + 33) \div 33$$

$$\text{Final Pressure (ATA)} = (\text{Final Depth} + 33) \div 33$$

3. Of the initial and final pressures (ATA), determine which is maximum and which is minimum:

Maximum Pressure (ATA) = Max(Initial Pressure, Final Pressure)

Minimum Pressure (ATA) = Min(Initial Pressure, Final Pressure)

4. Determine the maximum PO₂ and minimum PO₂ of the profile:

Maximum PO₂ = Maximum Pressure (ATA) × Fraction of O₂ in breathing gas

Minimum PO₂ = Minimum Pressure (ATA) × Fraction of O₂ in breathing gas

5. Determine if the maximum and minimum PO₂'s fall within the range for oxygen toxicity (i.e. greater than 0.5) and assign a new variable (Low PO₂) to describe the lowest PO₂ of the profile that is within the range for oxygen toxicity:

If Maximum PO₂ ≤ 0.5 then [calculation does not apply]

If Minimum PO₂ < 0.5 then Low PO₂ = 0.5, else Low PO₂ = Minimum PO₂

6. Determine time of exposure within range for oxygen toxicity:

$$\text{Exposure Time} = \text{Segment Time} \cdot \left(\frac{\text{Maximum PO}_2 - \text{Low PO}_2}{\text{Maximum PO}_2 - \text{Minimum PO}_2} \right)$$

Computer format:

In the FORTRAN programming language, the above steps can be written as:

```
SGTIME = (FDEPTH - SDEPTH)/RATE
IPATA = (SDEPTH + 33.0)/33.0
FPATA = (FDEPTH + 33.0)/33.0
MAXATA = MAX(IPATA, FPATA)
MINATA = MIN(IPATA, FPATA)
MAXPO2 = MAXATA * FO2 (GASMIX)
MINPO2 = MINATA * FO2 (GASMIX)
IF (MAXPO2 .LE. 0.5) GOTO 000 [exit this calculation sequence]
IF (MINPO2 .LT. 0.5) THEN
    LOWPO2 = 0.5
ELSE
    LOWPO2 = MINPO2
END IF
O2TIME = SGTIME*(MAXPO2 - LOWPO2)/(MAXPO2 - MINPO2)
```

Example Calculation:

Consider the following dive profile using EAN 32 for gasmix (ignore decompression considerations for this example):

- Segment 1 Descent from 0 fsw to 120 fsw at 40 fsw/min
- Segment 2 Constant depth at 120 fsw for 22 min
- Segment 3 Long, slow ascent from 120 fsw to 0 fsw at -4 fsw/min

How many OTU's are accumulated on this dive? Solution using computer format for equations:

For Segment 1

```
SGTIME = (120.0 - 0.0)/40.0 = 3.0 min
IPATA = (0.0 + 33.0)/33.0 = 1.0 ATA
FPATA = (120.0 + 33.0)/33.0 = 4.636 ATA
MAXATA = MAX(1.0, 4.636) = 4.636 ATA
MINATA = MIN(1.0, 4.636) = 1.0 ATA
MAXPO2 = 4.636*0.32 = 1.484 ATA
MINPO2 = 1.0*0.32 = 0.32 ATA
IF (1.484 .LE. 0.5) GOTO 000 [false, therefore continue]
IF (0.32 .LT. 0.5) THEN [true]
    LOWPO2 = 0.5
O2TIME = 3.0*(1.484 - 0.5)/(1.484 - 0.32) = 2.536 min
(using Equation 2)
OTU = 3.0/11.0*2.536/(1.484 - 0.5)*(((1.484 - 0.5)/0.5)
    ** (11.0/6.0) - ((0.5 - 0.5)/0.5)**(11.0/6.0))
= 2.431 OTU's
```

For Segment 2

```
PATA = (120.0 + 33.0)/33.0 = 4.636 ATA
PO2 = 4.636*0.32 = 1.484 ATA
(using Equation 1)
OTU = 22.0*((0.5/(1.484 - 0.5))**(-5.0/6.0)) = 38.664 OTU's
```

For Segment 3

```
SGTIME = (0.0 - 120.0)/-4.0 = 30.0 min
IPATA = (120.0 + 33.0)/33.0 = 4.636 ATA
FPATA = (0.0 + 33.0)/33.0 = 1.0 ATA
MAXATA = MAX(4.636, 1.0) = 4.636 ATA
MINATA = MIN(4.636, 1.0) = 1.0 ATA
MAXPO2 = 4.636*0.32 = 1.484 ATA
MINPO2 = 1.0*0.32 = 0.32 ATA
IF (1.484 .LE. 0.5) GOTO 000 [false, therefore continue]
IF (0.32 .LT. 0.5) THEN [true]
    LOWPO2 = 0.5
O2TIME = 30.0*(1.484 - 0.5)/(1.484 - 0.32) = 25.359 min
(using Equation 2)
OTU = 3.0/11.0*25.359/(0.5 - 1.484)*(((0.5 - 0.5)/0.5)
    ** (11.0/6.0) - ((1.484 - 0.5)/0.5)**(11.0/6.0))
= 24.309 OTU's
```

Total OTU's for this dive = 2.431 + 38.664 + 24.309 = **65.404 OTU's**

Central Nervous System (CNS) Oxygen Toxicity Calculations

Constant depth profile:

For a constant depth profile (in which the PO_2 remains constant), the CNS fraction is calculated by Equation 3:

$$\text{CNS fraction} = \frac{\text{time at } PO_2}{\text{time limit for } PO_2} \quad (\text{Eq. 3})$$

Ascent or descent profile at a constant rate:

For an ascent or descent profile at a constant rate (where the PO_2 varies at a constant rate), the CNS fraction for a given PO_2 range with a linear time limit function can be calculated by Equation 4:

$$\text{CNS fraction} = \frac{1}{mk} \cdot \left[\ln \left| T_{\text{lim}_i} + mkt_x \right| - \ln \left| T_{\text{lim}_i} \right| \right] \quad (\text{Eq. 4})$$

where m is the the slope of the linear time limit function, k is the constant rate of change in PO_2 with time, T_{lim_i} is the initial time limit for the interval, and t_x is the time of exposure over the interval.

Computer format:

In the FORTRAN programming language, Equation 3 can be written as:

```
CNS = TIME/TLIM
```

while Equation 4 can be written as:

```
CNS = 1.0/MK*(LOG (ABS (TLIMI + MK*TIME)) - LOG (ABS (TLIMI)))
```

Discussion:

As mentioned previously, it is convenient for calculation purposes to describe the NOAA CNS oxygen toxicity limits as a series of linear equations for various ranges of PO_2 . These equations express the time limit, T_{lim} , as a function of PO_2 in the slope-intercept form of $y = mx + b$. A straight line can be drawn through more than two NOAA time limit points for the PO_2 ranges from 0.9 to 1.1 and from 1.1 to 1.5. I have made one extrapolation to describe a time limit equation for the PO_2 range from 0.5 to 0.6 in order to complete the series. An individual equation is written as:

$$T_{\text{lim}} = mPO_2 + b \quad (\text{Eq. 5})$$

where m is the slope of the line and b is the intercept for the given PO₂ range. Coefficients for the series of linear equations are as follows:

<u>PO₂ Range</u>	<u>Slope</u>	<u>Intercept</u>
0.5 – 0.6	–1800	1800
0.6 – 0.7	–1500	1620
0.7 – 0.8	–1200	1410
0.8 – 0.9	–900	1170
0.9 – 1.1	–600	900
1.1 – 1.5	–300	570
1.5 – 1.6	–750	1245

Calculation of T_{lim} within a PO₂ range applies when the present PO₂ is greater than the lower limit of the PO₂ range and is less than or equal to the upper limit of the PO₂ range.

Equation 3 can now be expressed as:

$$\text{CNS fraction} = \frac{\text{time at PO}_2}{T_{\text{lim}}} = \frac{\text{time at PO}_2}{m\text{PO}_2 + b} \quad (\text{Eq. 6})$$

Equation 4 can be derived from Equation 6 as follows:

Since the PO₂ varies at a constant rate with respect to time during an ascent or descent at a constant rate, then

$$\frac{d\text{PO}_2}{dt} = k \quad \int_{\text{PO}_{2_0}}^{\text{PO}_{2_f}} d\text{PO}_2 = \int_{t_0}^{t_f} k dt \quad \text{PO}_{2_f} = \text{PO}_{2_0} + kt$$

Equation 6 can now be expressed as a differential with PO₂ as a function of time:

$$d\text{CNS fraction} = \frac{dt}{m(\text{PO}_{2_0} + kt) + b} = \frac{dt}{m\text{PO}_{2_0} + mkt + b}$$

Setting $a = mPO_2_o + b$ ($= T_{lim_i}$) and $b = mk$, Equation 4 is obtained from the following integration:

$$\int_{CNS_o}^{CNS_f} dCNS \text{ fraction} = \int_{t_o}^{t_f} \frac{dt}{a + bt}$$

Computer programming method to access linear equations for T_{lim} :

A method to store and “look up” the PO_2 ranges and coefficients for the linear time limit equations must be provided in a program. In the FORTRAN programming language, this can be accomplished in a BLOCK DATA subprogram using one-dimensional arrays with subscripted variables:

```
BLOCK DATA CNSLIM
REAL PO2LO, PO2HI, LIMSLP, LIMINT
DIMENSION PO2LO(7), PO2HI(7), LIMSLP(7), LIMINT(7)
COMMON /A/ PO2LO, PO2HI, LIMSLP, LIMINT
DATA PO2LO(1)/0.5/, PO2LO(2)/0.6/, PO2LO(3)/0.7/, PO2LO(4)/0.8/,
* PO2LO(5)/0.9/, PO2LO(6)/1.1/, PO2LO(7)/1.5/
DATA PO2HI(1)/0.6/, PO2HI(2)/0.7/, PO2HI(3)/0.8/, PO2HI(4)/0.9/,
* PO2HI(5)/1.1/, PO2HI(6)/1.5/, PO2HI(7)/1.6/
DATA LIMSLP(1)/-1800.0/, LIMSLP(2)/-1500.0/, LIMSLP(3)/-1200.0/,
* LIMSLP(4)/-900.0/, LIMSLP(5)/-600.0/, LIMSLP(6)/-300.0/,
* LIMSLP(7)/-750.0/
DATA LIMINT(1)/1800.0/, LIMINT(2)/1620.0/, LIMINT(3)/1410.0/,
* LIMINT(4)/1170.0/, LIMINT(5)/900.0/, LIMINT(6)/570.0/,
* LIMINT(7)/1245.0/
END
```

where

PO2LO = low end of PO_2 range

PO2HI = high end of PO_2 range

LIMSLP = slope coefficient for PO_2 range

LIMINT = intercept coefficient for PO_2 range

Calculation steps for constant depth profile:

1. Determine if PO_2 is within the range for oxygen toxicity:

If $PO_2 \leq 0.5$ then [calculation does not apply]
2. Find which PO_2 range the present PO_2 falls within in order to calculate T_{lim} . This step is accomplished by use of a program loop with the provision to exit the loop once the correct PO_2 range is found.
3. Calculate the time limit, T_{lim} , for the present PO_2 .
4. Calculate the CNS fraction using Equation 3.

Computer format:

In the FORTRAN programming language, the above steps can be written as:

```
      IF (PO2 .LE. 0.5) GOTO 000 [exit this calculation sequence]
      DO 10 I = 1,7
      IF ((PO2 .GT. PO2LO(I)) .AND. (PO2 .LE. PO2HI(I)) THEN
          TLIM = LIMSLP(I)*PO2 + LIMINT(I)
          GOTO 000 [exit loop]
      END IF
10    CONTINUE
      CNS = TIME/TLIM
```

Calculation steps for ascent or descent profile at a constant rate:

The first six (6) steps in this sequence are identical to the OTU calculation for ascent or descent profile. The additional steps required are:

7. Determine how much of the oxygen exposure time is spent within each PO₂ range (each of which has different CNS time limit equation coefficients) and determine what the initial PO₂ and final PO₂ values are within each respective range. This step is accomplished using structured IF blocks within a program loop and assigns the oxygen exposure time, initial PO₂, and final PO₂ for each range to subscripted variables in arrays. The oxygen exposure time for a PO₂ range is given by: $O_2 \text{ time for range} = O_2 \text{ time for segment} \times \text{absolute value} (\text{final PO}_2 \text{ for range} - \text{initial PO}_2 \text{ for range}) \div (\text{max PO}_2 \text{ for segment} - \text{low PO}_2 \text{ for segment})$.
8. Calculate the initial time limit and product of constants, $m \times k$, for each PO₂ range. Calculate the CNS fraction for each PO₂ range using Equation 4. This step is accomplished with a program loop and skips the calculation for a particular PO₂ range if there was no oxygen exposure time within that range.
9. Sum up the CNS fractions for each PO₂ range to produce a CNS fraction for the dive segment. This step is accomplished with a program loop.

Computer format:

In the FORTRAN programming language, the above steps can be written as:

```
SGTIME = (FDEPTH - SDEPTH)/RATE
IPATA = (SDEPTH + 33.0)/33.0
FPATA = (FDEPTH + 33.0)/33.0
MAXATA = MAX(IPATA, FPATA)
MINATA = MIN(IPATA, FPATA)
MAXPO2 = MAXATA * FO2(GASMIX)
MINPO2 = MINATA * FO2(GASMIX)
IF (MAXPO2 .LE. 0.5) GOTO 000 [exit this calculation sequence]
IF (MINPO2 .LT. 0.5) THEN
    LOWPO2 = 0.5
ELSE
    LOWPO2 = MINPO2
END IF
O2TIME = SGTIME*(MAXPO2 - LOWPO2)/(MAXPO2 - MINPO2)
```

```

DO 10 I = 1,7
IF ((MAXPO2 .GT. PO2LO(I)) .AND. (LOWPO2 .LE. PO2HI(I))) THEN
  IF ((MAXPO2 .GE. PO2HI(I)) .AND. (LOWPO2 .LT. PO2LO(I))) THEN
    IF (SDEPTH .GT. FDEPTH) THEN
      PO2O(I) = PO2HI(I)
      PO2F(I) = PO2LO(I)
    ELSE
      PO2O(I) = PO2LO(I)
      PO2F(I) = PO2HI(I)
    END IF
    SEGPO2(I) = PO2F(I) - PO2O(I)
  ELSE IF ((MAXPO2 .LT. PO2HI(I)) .AND. (LOWPO2 .LE. PO2LO(I))) THEN
    IF (SDEPTH .GT. FDEPTH) THEN
      PO2O(I) = MAXPO2
      PO2F(I) = PO2LO(I)
    ELSE
      PO2O(I) = PO2LO(I)
      PO2F(I) = MAXPO2
    END IF
    SEGPO2(I) = PO2F(I) - PO2O(I)
  ELSE IF ((LOWPO2 .GT. PO2LO(I)) .AND. (MAXPO2 .GE. PO2HI(I))) THEN
    IF (SDEPTH .GT. FDEPTH) THEN
      PO2O(I) = PO2HI(I)
      PO2F(I) = LOWPO2
    ELSE
      PO2O(I) = LOWPO2
      PO2F(I) = PO2HI(I)
    END IF
    SEGPO2(I) = PO2F(I) - PO2O(I)
  ELSE
    IF (SDEPTH .GT. FDEPTH) THEN
      PO2O(I) = MAXPO2
      PO2F(I) = LOWPO2
    ELSE
      PO2O(I) = LOWPO2
      PO2F(I) = MAXPO2
    END IF
    SEGPO2(I) = PO2F(I) - PO2O(I)
  END IF
  OTIME(I) = O2TIME*ABS(SEGPO2(I))/(MAXPO2 - LOWPO2)
ELSE
  OTIME(I) = 0.0
END IF
10 CONTINUE
DO 20 I = 1,7
IF (OTIME(I) .EQ. 0.0) THEN
  CNS(I) = 0.0
  GOTO 20
ELSE
  TLIMI(I) = LIMSLP(I)*PO2O(I) + LIMINT(I)
  MK(I) = LIMSLP(I)*(SEGPO2(I)/OTIME(I))
  CNS(I) = 1.0/MK(I)*(LOG(ABS(TLIMI(I) + MK(I)*OTIME(I))) -
* LOG(ABS(TLIMI(I))))
END IF
20 CONTINUE
SUMCNS = 0.0
DO 30 I = 1,7
  TMPCNS = SUMCNS
  SUMCNS = TMPCNS + CNS(I)
30 CONTINUE

```

Example Calculation:

Consider the following dive profile using EAN 32 for gasmix (ignore decompression considerations for this example):

Segment 1 Descent from 0 fsw to 120 fsw at 40 fsw/min
Segment 2 Constant depth at 120 fsw for 22 min
Segment 3 Long, slow ascent from 120 fsw to 0 fsw at -4 fsw/min

What is the CNS fraction accumulated on this dive? Program solution:

For Segment 1 Program calculates a CNS fraction = .0090

For Segment 2 Program calculates a CNS fraction = .1761

For Segment 3 Program calculates a CNS fraction = .0895

Total CNS fraction for this dive = .0090 + .1761 + .0895 = **.2746 = 27.5%**

Conclusion

The information presented in this article is intended to suggest some precise methods to calculate oxygen toxicity parameters within the framework of their present usage in technical diving. I do not intend to suggest or imply any new physiological insight as the result of some mathematical manipulations. The idea is simply to make accurate calculations of oxygen toxicity as a function of the actual oxygen partial pressure (PO_2) and the actual time of exposure.

References / More Information

Clark, J.M. Oxygen Toxicity. In *The Physiology and Medicine of Diving*, pp. 121-169. Eds. P.B. Bennett & D.H. Elliott. 4th ed. London; Philadelphia: W.B. Saunders, 1993.

Hamilton, R.W., Kenyon, D.J., Peterson, R.E., Butler, G.J., & Beers, D.M. *REPEX: Development of Repetitive Excursions, Surfacing Techniques, and Oxygen Procedures for Habitat Diving*. NURP Technical Report 88-1A, 1988.

Hamilton, R.W., Kenyon, D.J. & Peterson, R.E. *REPEX Habitat Diving Procedures: Repetitive Vertical Excursions, Oxygen Limits, and Surfacing Techniques*. NURP Technical Report 88-1B, 1988.

Lambertsen, C.J. Extension of Oxygen Tolerance in Man: Philosophy and Significance. *Experimental Lung Research*. **14** (Suppl.), pp. 1035-1058, 1988.

NOAA Diving Manual, US Department of Commerce, October 1991.